

UnlightVBE-QS 腳本開發說明書

2022/5/10

Ver. 1.62

目錄

1. 前言.....	1
2. 腳本運作流程介紹	2
3. 執行階段.....	3
I. 運作簡介與時間軸圖	3
II. 執行階段型態分類.....	6
III. 執行階段使用面向.....	6
IV. 執行階段詳細說明.....	7
普通型.....	7
事件型.....	9
特殊型.....	10
V. 啟動管道與發動順序	11
VI. 執行週期與待命執行	12
4. 執行指令集	13
I. 一般指令	14
啟動類.....	14
系統類.....	15
移動距離類.....	16
骰數控制類.....	17
人物類.....	18
卡牌類.....	20
通用事件回應類	22
II. 附加狀態	23

人物實際狀態類	23
異常狀態類	24
III. 其他事項	26
5. 腳本引數變數	27
6. 腳本體系類型	33
7. 腳本及卡片人物資訊文件格式.....	35
I. 卡片人物資訊 (UnlightVBE Card Information)	35
II. 技能腳本文件	40
8. 主動技能之 AI 計算判斷設計	49
9. 腳本之錯誤處理	54
10. 附錄 1-人物角色之圖片資源概要	59
11. 附錄 2-人物角色之小人物立繪影子座標取得操作	61

1. 前言

UnlightVBE 經歷了 2 個正式版本的發布 (Version α 、 Version ζ) , 雖然其遊戲系統完整性日益成熟, 不過在人物角色的資訊的自由編輯及可攜性方面, 仍然是相當的有限。在追求人物角色技能程式碼外部化, 達到角色資訊完全獨立的期望之下, 本版本 (QS) 也就此誕生。


在當初開始作 UnlightVBE 時, 其實最先是抱持著一種興致使然, 一切都是原始於當時的一個小巧簡單構造的程式, 而這股純粹的興致讓我不知不覺一邊在學習 VB 之下, 另一邊在慢慢修改著程式, 不知什麼時候它竟變得如此巨大, 到最後還真的搞出一個可以玩的遊戲程式出來, 就連現在的我回想當時之種種回憶也是很難以想像。

本說明書旨在說明本遊戲系統對於腳本之運作, 以及說明腳本之架構與指令撰寫, 本人也希望透過本遊戲平台及開發說明書, 能夠讓更多二次元之人物角色有登上戰鬥舞台的機會。

UnlightVBE (QS) 製作者 Andy Ciu.

2. 腳本運作流程介紹

本腳本文件採用 VBScript 語言撰寫，搭配 VB6 之 Microsoft Script Control 1.0 之套件來讀取及執行腳本相關程式碼，透過在遊戲中特定之時間點觸發腳本文件，由其所回傳之指令字串來詮釋技能之效果。



UnlightVBE-QS 主要是利用 2 個子系統來實作：執行階段系統&執行指令集系統，在遊戲戰鬥過程中，依照時間順序及事件劃分成若干個「執行階段」，當階段觸發時，執行階段系統負責統計戰鬥資訊及先後順序，隨後執行該角色之腳本文件，並接收不等的「執行指令」字串，再透過執行指令集系統來依照個別指令去執行特定的效果。每個角色，乃至於每個必殺技/狀態都為一獨立之文件個體，可實現理論上若干多個角色之實裝而不受其遊戲主程式之影響與綁定限制。

3. 執行階段





I. 運作簡介與時間軸圖





本遊戲對於腳本之執行採用時間軸型態的編輯模式，對每個關鍵時間點設置執行階段，來達到「在什麼時間做什麼事」之目的。

在戰鬥的過程中，有既定之戰鬥流程與規則，對本遊戲來說，能夠在適當的時間中打出最完美的牌，讓角色使用各種的必殺技，發揮出最大的戰鬥能力，是玩家所追求之首要。

作為一位人物角色之設計者，在設計技能能力及數值之時，對於卡牌相對應的條件，對其他人物角色之對應關係尤其重要，這也牽涉到戰鬥過程與時間先後順序之流程應對。在甚麼樣的情況下某些特定之技能能夠發揮最大的效益，又在什麼情況下這些技能有可能遭受到箝制，甚至是無法顯現出原本應有的功能，這些事項想必設計者都會去再三思考吧。

為充分展現戰鬥流程之各種事件與應對關係，本系統採用時間軸制之事件列舉，並把設計者可用之時間點作編號，每一個時間點以一個「執行階段」作為稱呼。為方便設計者掌握各個執行階段點之時間先後順序，以下就以時間軸之概念敘述系統事件、執行階段點之時間關係：





II. 執行階段型態分類

在執行階段系統中，執行階段分為以下 3 種類型：

A. 普通型 - 在時間軸圖上之執行階段點：

發牌階段：0,1

移動階段：2~9,70,71

攻擊階段：10~17

防禦階段：30~37

攻擊/防禦階段最後執行點：20~29

回合結束階段：50~55

B. 事件型 - 在遊戲中某個時間點由事件所觸發。

C. 特殊型 - 在遊戲中屬於特殊類型之執行階段點。

若是您一開始不知該如何下手的話也沒關係，一般以主動技能來說，首先需要從出牌條件是否成立開始寫起，這時會利用到特殊型之相關出牌事件來做啟動之驗證，再來就是發動技能的執行效果的部分了，這部分就是依照技能來自由運用其事件，像是您可以利用時間軸上各個普通型執行階段事件來發動技能動畫，動畫執行中您可以利用 (61) 來執行技能效果，或者您也可以運用事件型執行階段事件來對戰鬥中某些特定的時間點做出相符合的動作等等。

III. 執行階段使用面向

在普通型執行階段中，通常都會依照玩家雙方以及各個不同的技能體系來逐一進行廣播，讓有利用到此執行階段之各項技能腳本都能進行參與，不過像事件型與特殊型的執行階段則會因為其事件之需要而限制其使用面向範圍，像是該事件只能由我方參與，敵方不能參與之類的一些限制。以下表格為針對事件型、特殊型執行階段事件的使用面向範圍名詞解釋說明：

使用面向範圍

雙方	該事件玩家雙方都可參與
單方	該事件只有發動此事件之一方才可參與
全域	該事件可參與方的全部技能體系都可參與
個體	該事件可參與方只有發動此事件之特定技能體系才可參與

舉例來說，「雙方 / 全域」代表玩家雙方所有技能體系都能夠利用此執行階段事件，而「單方 / 個體」則只有發動此執行階段事件之特定一方的特定技能體系才能夠使用此事件。

IV. 執行階段詳細說明

普通型

0/1	主要對被動技-發牌階段效果實施	發牌
2/3/4	移動動作執行前之主動技效果發動實施	移動
70/71	移動動作執行前之異常狀態效果實施	移動
5/6/7	移動階段主要之效果實施	移動
8	針對於 HP 存活檢查前之效果實施	移動
9	針對於交換角色後之效果實施	移動
10	針對具優先權之技能效果實施 (EX: 主動技-因果之輪)	攻擊
11	攻擊階段主要之效果實施	攻擊
12	針對於雙方總骰數決定後之效果實施	攻擊
13	主要針對需要進行多次擲骰之技能效果實施	攻擊
14/15/16	在骰傷傷害執行後之技能效果實施 (EX:主動技-因果之刻)	攻擊
17	主要對被動技-攻擊階段效果實施	攻擊
30	針對具優先權之技能效果實施	防禦

31	防禦階段主要之效果實施	防禦
32	針對於雙方總骰數決定後之效果實施	防禦
33	主要針對需要進行多次擲骰之技能效果實施	防禦
34/35/36	在骰傷傷害執行後之技能效果實施	防禦
37	主要對被動技-防禦階段效果實施	防禦
20	擲骰後效果實施-一般類型	骰後執行點
21	擲骰後效果實施-追加/減輕骰數類型	骰後執行點
22	擲骰後效果實施-其他骰數有關類型	骰後執行點
23	擲骰後效果實施-攻擊/防禦成功類型	骰後執行點
24	擲骰後效果實施-骰數乘數加成類型 (EX:異常狀態-恐怖・異常狀態-狂戰士)	骰後執行點
25	擲骰後效果實施-骰數有關類型-防禦 (EX:主動技-茨林)	骰後執行點
26	擲骰後效果實施-卡牌抽牌/回牌類型 (EX:主動技-反擊的狼煙)	骰後執行點
27	擲骰後效果實施-骰數特殊類型 (EX:異常狀態-不死)	骰後執行點
28	擲骰後效果實施-卡牌丟牌類型 (EX:主動技-雷擊)	骰後執行點
29	擲骰後效果實施-骰傷傷害類型 (EX:主動技-兩個身體・主動技-不屈之心)	骰後執行點
50/51/52	針對於 HP 存活檢查前之效果實施	回合結束
53/54/55	回合結束階段之主要效果實施 / 結束執行	回合結束

事件型

41	角色交換時	雙方 / 全域
人物角色交換時觸發事件。		
46	人物角色受到傷害時	雙方 / 全域
人物角色受到傷害時觸發事件，包含毆傷 / 直傷 / 直接死亡。		
47	人物角色移動時	雙方 / 全域
人物角色移動時觸發事件。		
48	人物角色回復血量時	雙方 / 全域
人物角色回復血量時觸發事件。		
49	人物角色復活時	雙方 / 全域
人物角色復活時觸發事件。		
61	動畫發生時途中之效果實行	單方 / 個體
技能動畫發動時觸發之被動事件。		
62	技能效果進行多次擲骰時	單方 / 個體
技能效果進行多次擲骰時觸發之被動事件，每次擲骰後須在本執行階段手動進行結果紀錄。		
72	異常狀態新增時	單方 / 個體
異常狀態新增時觸發之被動事件， 異常狀態專用 。		
73	異常狀態消滅時	單方 / 個體
異常狀態消滅時觸發之被動事件， 異常狀態專用 。		
74	人物實際狀態新增時	單方 / 個體
人物實際狀態新增時觸發之被動事件， 人物實際狀態專用 。		
75	人物實際狀態消滅時	單方 / 個體
人物實際狀態消滅時觸發之被動事件， 人物實際狀態專用 。		
76	人物角色附加狀態增加時	雙方 / 全域
人物角色附加狀態增加時觸發事件，於執行階段 72/74 後執行。		
77	人物角色附加狀態解除時	雙方 / 全域
人物角色附加狀態解除時觸發事件，於附加狀態消滅後執行。		

特殊型

42 攻擊階段-出牌時 單方 / 全域

玩家出牌時觸發事件（攻擊階段），主要為主動技能進行發動之驗證。

註：未啟動之技能可使用此事件，但待命角色之主動技能無論是否已啟動皆除外。

43 防禦階段-出牌時 單方 / 全域

玩家出牌時觸發事件（防禦階段），主要為主動技能進行發動之驗證。

註：未啟動之技能可使用此事件，但待命角色之主動技能無論是否已啟動皆除外。

44 移動階段-出牌時 雙方 / 全域

玩家出牌時觸發事件（移動階段），主要為主動技能進行發動之驗證。

註：未啟動之技能可使用此事件，但待命角色之主動技能無論是否已啟動皆除外。

45 雙方總骰數更新時 雙方 / 全域

雙方玩家總骰數更新時觸發事件，主要為一般效果骰數加成時之統一管道。

92 攻擊階段-階段開始時 單方 / 全域

雙方玩家出牌前系統事件（攻擊階段），主要為技能發動之驗證（例如：無卡牌要件者）。

註¹：未啟動之技能可使用此事件。

註²：本事件於普通型執行階段 17 後執行。

93 防禦階段-階段開始時 單方 / 全域

雙方玩家出牌前系統事件（防禦階段），主要為技能發動之驗證（例如：無卡牌要件者）。

註¹：未啟動之技能可使用此事件。

註²：本事件於普通型執行階段 37 後執行。

94 移動階段-階段開始時 雙方 / 全域

雙方玩家出牌前系統事件（移動階段），主要為技能發動之驗證（例如：無卡牌要件者）。

註¹：未啟動之技能可使用此事件。

註²：本事件於普通型執行階段 1 後執行。

99 智能判斷型 AI 計算卡牌之排列組合時 單方 / 個體

智能判斷型 AI 計算卡牌之排列組合時觸發事件，針對 AI 於徵求主動技能之意見時所使用。

V. 啟動管道與發動順序

不論是哪一型態的執行階段點，都需要先對系統進行「啟動」確認之後，方可進行使用。「啟動」分為 2 個部分，第 1 個部分是該技能腳本於執行階段點的登記，可參照第 7 章（腳本及卡片人物資訊文件格式-技能腳本文件-初始架構標頭資訊說明）。第 2 個部分則是在遊戲過程中，該技能是否處在發動中狀態。以下表格說明各執行階段型態中系統如何驗證其為發動中狀態：

第 2 部分啟動管道確認

執行階段型態	說明/備註
普通型	主動/被動技能體系： 已發送 <u>AtkingTurnOnOff</u> 指令為開啟者 (唯有在啟動後才能在其後的各個執行階段點進行使用)
	異常 / 人物實際技能體系： 已附著在人物角色者
事件型	僅具有觸發對應之 <u>發動中</u> 技能體系才具使用資格
特殊型	詳見（執行階段詳細說明-特殊型）之各執行階段說明

各技能體系之發送啟動指令時機

體系	說明
主動技能	於特殊型執行階段 <u>42/43/44/92/93/94</u> 發送
被動技能	除主動技能之管道外，可於任何 <u>普通型</u> 執行階段發送
異常狀態	不須主動發送（狀態存續即在發動中狀態）
人物實際狀態	

在系統判斷是否啟動確認的部分，於執行階段系統會先做一次檢查，而後執行指令集系統會針對主/被動技能，在每次各執行指令解釋前都會再行驗證是否處在發動中狀態，若無者則不予解釋執行除（AtkingLineLight/ AtkingTurnOnOff）之外的其他指令。

發動順序部分，普通型執行階段在啟動之後，因雙方都具有權力來發動各執行階段點，系統在不同時間下會照不同的規則來作為執行，以下表格為各階段之發動順序：

普通型各執行階段點之發動順序

執行階段號	發動順序
0~4/70/71	使用者方先發動
5~9	先搶得主動權之一方則先發動
10~17	攻擊方先發動
30~37	防禦方後發動
20~29	攻擊方先發動
50~55	有主動權之一方則後發動

至於事件型與特殊型則一律以發動事件的那一方為主，而後再依照其使用面向範圍再決定是否允許敵方參與發動。

VI. 執行週期與待命執行

通常技能以一個回合當作開始與結束之執行週期，但由於原則上技能之開關與否都交由技能體系自行控制，故理論上能實現所謂跨執行週期（跨回合數）之寫法。

但要注意的一點是，在效果執行前務必要釐清自身角色於場上之狀態定位，亦及該角色是否為場上角色或為待命中角色。在被動技能 / 異常狀態 / 人物實際狀態中由於其特性之緣故，故都採取先行判斷角色場上狀態之邏輯寫法，而主動技能在一般情況下因不允許在待命角色中執行之關係，故邏輯判斷上可單純化無須介意，但若設計者開啟了主動技能可在待命角色中執行之功能，則應當在每個使用之執行階段注意其自身角色之場上狀態。

4. 執行指令集

本系統是以腳本回傳之「指令字串」作為效果之執行依據。指令字串是由數個指令串連而成，透過系統指定之指令格式，來達到效果詮釋之目的。

因應每個執行指令其使用範圍之不同，之後於每項指令右方都有其特定標記，說明如下：

特定標記圖例	
主動技能	指定之技能體系可使用
異常狀態	指定之技能體系 排除 使用
普通型	指定之執行階段型態或特定範圍可使用
13/33	指定之執行階段型態或特定範圍 排除 使用
61	該指令所觸發對應之執行階段事件（標記*者需參考備註）

若無技能體系特定標記者，則屬全體適用。

以下表格為各執行指令之詳細說明：

I. 一般指令

啟動類				
AtkingLineLight	#n1	主動技能	被動技能	45/99
人物必殺技狀態燈控制(自身) (註：本指令有效範圍僅限於場上角色。) n1-(1.開啟/2.關閉)				
AtkingTurnOnOff	#n1	主動技能	被動技能	45/99
人物必殺技啟動碼控制(自身) n1-(1.開啟/2.關閉)				
AtkingLineLightAnother	#n1,n2,n3	主動技能	被動技能	普通型 事件型
人物必殺技狀態燈控制(其他) (註 ¹ ：本指令有效範圍僅限於場上角色。) (註 ² ：本指令僅限發動中之技能才能夠使用。) n1-(1.主動技能/2.被動技能) n2-(第 x 個技能) n3-(1.開啟/2.關閉)				
AtkingTurnOnOffAnother	#n1,n2,n3	主動技能	被動技能	普通型 事件型
人物必殺技啟動碼控制(其他) (註：本指令僅限發動中之技能才能夠使用。) n1-(1.主動技能/2.被動技能) n2-(第 x 個技能) n3-(1.開啟/2.關閉)				

系統類				
AtkingStartPlay	#n1,n2	61	普通型	異常狀態 13/33
技能動畫圖像執行 (靜態)				
n1-(第 1 張圖片路徑) n2-(第 2 張圖片路徑(無則填 0))				
AtkingStartPlayAnimate	#n1,[n2,n3]	61	普通型	異常狀態 13/33
技能動畫圖像執行 (逐格動畫) [實驗性功能]				
(註 ¹ : 可擺放 16~24 張圖片，檔案以 1.png/2.png/...編號逐一命名)				
(註 ² : 圖片資料夾路徑字串結尾須加反斜線(\))				
(註 ³ : 可選擇關鍵影格 1~2 張作為靜態技能動畫圖像執行使用，適用於在設定中關閉逐格動畫支援者；若無填寫時則系統預設取第 4 張與第 12 張影格作為使用)				
n1-(圖片資料夾路徑) [n2]-(關鍵影格第 1 張圖片編號) [n3]-(關鍵影格第 2 張圖片編號)				
BattleTurnControl	#n1,n2		普通型	事件型
系統回合數控制				
n1-(1.加/2.減) n2-(數值)				
BattleSendMessage	#n1			99
發布訊息效果執行				
n1- (訊息內容)				
BattleStartDice	#0	62	13/33/20/21/22/23/24/25/26/27/28/29	
擲骰效果執行				
(註 : 如需進行多次擲骰，則需重複執行多次本指令。另外在每次擲骰之後，會觸發執行階段-62 事件，技能需在本執行階段進行擲骰之結果紀錄，系統不予保留。)				
AtkingInformationRecord	#n1		主動技能	被動技能 99
技能自行註記之備註資訊記載				
(註 : 本備註資訊會更新在引數 AtkingckVSF 中。)				
n1-(技能自身註記之備註字串)				

EventPlayerAllActionOff #n1

1/17/37

禁止玩家進行所有操作控制

(註¹：本指令於玩家各行動階段之第 1 個執行階段進行效果，例如欲在該玩家攻擊階段實施者，則在執行階段 17 下達本指令，依此類推。)

(註²：本指令執行效果僅以當下行動階段為限。)

n1-(1.自身/2.對方)

移動距離類

BattleMoveControl

#n1

47

普通型

事件型

角色相對距離控制

n1-(1.近距離/2.中距離/3.遠距離)

PersonMoveControl

#n1,n2,n3

2/3/4/70

人物移動階段總移動量控制

(註：一旦進行「指定」數值之後，即無視各指令原讓其增加或減少之變化量)

n1-(1.自身/2.對方)

n2-(1.增加/2.減少/3.指定)

n3-(數值)

PersonMoveActionChange #n1,n2

2/3/4/70

人物角色移動階段行動控制

(註：在 1v1 戰鬥模式下更換角色指令為無效果。)

n1-(1.自身/2.對方)

n2-(1:向遠/2:回復/3:向近/4:更換角色/5:不選擇)

PersonAttackFirstControl #n1

2/3/4/70/71

人物角色優先攻擊控制

(註¹：一旦指定後即無視移動總量對於先後攻之判定。)

(註²：若指定自身先攻即是讓對方後攻，反之亦然。)

n1-(1.自身/2.對方)

骰數控制類

EventTotalDiceChange #n1,n2,n3

45

攻擊/防禦階段系統骰數變化量控制

(註¹：本指令為攻擊/防禦階段「出牌時」使用，適用一般對骰量加成之效果)

(註²：一旦指定數值之後，即無視在本指令下各增加或減少之控制)

n1-(1.自身/2.對方)

n2-(1.加/2.減/3.乘/4.除(尾數捨去)/5.除(尾數進位)/6.指定)

n3-(數值)

EventPersonAbilityDiceChange #n1,n2,n3

45

攻擊/防禦階段角色白值能力對骰數變化量控制

(註¹：本指令對骰量加成之效果是在 EventTotalDiceChange 之前計算)

(註²：一旦指定數值之後，即無視在本指令下各增加或減少之控制，亦無視角色原本之基礎白值設定)

n1-(1.自身/2.對方)

n2-(1.加/2.減/3.指定)

n3-(數值)

PersonTotalDiceControl #n1,n2,n3

10/11/30/31

攻擊/防禦階段擲骰前系統總骰數直接控制

(註：本指令為攻擊/防禦階段於雙方按下 OK 後，擲骰之前使用，適用於對總骰量具影響之效果)

n1-(1.自身/2.對方)

n2-(1.加/2.減/3.乘/4.除(尾數捨去)/5.除(尾數進位)/6.指定)

n3-(數值)

AtkingTrueDiceControl #n1,n2

20/21/22/23/24/25/26/27/28/29

擲骰後正面骰數控制

(註：本指令是在擲骰後，針對雙方之正面骰數差數作控制，例如我方正面骰 5，對手正面骰 2，則在我方為攻擊階段時，系統正面骰數為 3，在我方為防禦階段時，系統正面骰數為 -3。)

n1-(1.加成/2.減少/3.指定)

n2-(數值)

人物類

PersonBloodControl #n1,n2,n3,n4 46/48* 普通型 事件型 46/48

人物角色血量控制

(註：執行本指令會觸發執行階段-46[直接傷害/直接死亡]或 48[回復]事件)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(1.直接傷害/2.回復/3.直接死亡)

n4-(數值)

PersonAtkingOff #n1,n2,n3 異常狀態 人物實際狀態 普通型 事件型

禁止人物主動技能發動效果控制(整體)

(註：在本指令標記開啟後，除已啟動之主動技外均無法在任何執行階段執行，直到再次使用本指令標記關閉後方可啟動。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(1.開啟/2.關閉)

PersonPassiveOff #n1,n2,n3 異常狀態 人物實際狀態 普通型 事件型

禁止人物被動技能發動效果控制(整體)

(註：在本指令標記開啟後，除已啟動之被動技外均無法在任何執行階段執行，直到再次使用本指令標記關閉後方可啟動。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(1.開啟/2.關閉)

PersonAtkingOffSelect #n1,n2,n3,n4 異常狀態 人物實際狀態 普通型 事件型

禁止人物主動技能發動效果控制(選擇)

(註¹：在本指令標記開啟後，除非該技能仍在啟動中，否則均無法在任何執行階段執行，直到再次使用本指令標記關閉後方可啟動。)

(註²：使用 PersonAtkingOff 指令時會複寫本指令之設定。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(1.開啟/2.關閉)

n4-(第 x 個主動技能)

PersonPassiveOffSelect #n1,n2,n3,n4

異常狀態

人物實際狀態

普通型

事件型

禁止人物被動技能發動效果控制(選擇)

(註¹：在本指令標記開啟後，除非該技能仍在啟動中，否則均無法在任何執行階段執行，直到再次使用本指令標記關閉後方可啟動。)

(註²：使用 PersonPassiveOff 指令時會複寫本指令之設定。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(1.開啟/2.關閉)

n4-(第 x 個被動技能)

PersonAtkingInvalid #n1,n2

普通型

事件型

人物必殺技無效化

(註：本指令只將技能強制關閉並解除對骰量造成之變化量，並無法回溯在發動本指令之前該技能對戰況之影響。)

n1-(1.自身/2.對方)

n2-(1.主動技/2.被動技)

PersonResurrect #n1,n2

49

普通型

人物角色復活效果執行

(註¹：若發動指令時當下該指定之角色 HP 大於 0 者則無效。)

(註²：利用本指令復活後之角色 HP 為 1，但其卡牌最大數量上限並不會隨之減少。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

PersonChangeBattleImage #n1,n2

主動技能

被動技能

普通型

事件型

人物角色更換戰鬥擲骰立繪

(註：本指令僅一次性覆寫目標人物之立繪資料，並非為狀態之儲存。)

n1-(1.自身/2.對方)

n2-(欲套用之大人物形象圖片路徑)

卡牌類

AtkingSeizeEnemyCards	#n1,n2	普通型	事件型
對手卡牌奪取效果執行			
n1-(1.手牌類/2.出牌類)			
n2-(牌編號(VBEPageCardNumVS 順序))			
AtkingDrawCards	#n1,n2,n3	普通型	事件型
抽牌效果執行			
(註：抽取事件卡時會直接推進回合數，1v1 時符合規則下第 13 回合後不會再抽取事件卡。)			
n1-(1.自身/2.對方)			
n2-(1.公用牌類/2.事件卡類)			
n3-(抽牌張數)			
BattleDeckShuffle	#0	普通型	事件型
強制洗牌執行			
AtkingDestroyCards	#n1,n2	普通型	事件型
丟牌效果執行			
n1-(1.自身/2.對方)			
n2-(牌編號(VBEPageCardNumVS 順序))			
AtkingGiveCards	#n1	普通型	事件型
自身卡牌送與效果執行			
n1-(牌編號(VBEPageCardNumVS 順序))			
AtkingGetUsedCards	#n1	普通型	事件型
墓地牌回牌效果執行			
n1-(牌編號(VBEPageCardNumVS 順序))			
AtkingOneSelfCardControl	#n1,n2,n3	2/3/4/70/10/11/12/17/30/31/32/37	
對某一方擁有之卡牌進行控制			
n1-(1.自身/2.對方)			
n2-(1.手牌出牌/2.出牌回牌/3.轉牌)			
n3-(牌編號(VBEPageCardNumVS 順序))			

PersonMaxCardsNumControl #n1,n2,n3

普通型

事件型

對人物角色之卡牌最大數量上限控制

n1-(1.自身/2.對方)

n2-(1.增加/2.減少)

n3-(數值)

BattleInsertEventCard #n1,n2,n3

普通型

事件型

插入事件卡效果執行

(註：事件卡標題為系統指定之字串，詳細請參照本章第 3 節「其他事項」之說明。)

n1-(1.自身/2.對方)

n2-(第 x 張(戰鬥回合數順序))

n3-(事件卡標題)

通用事件回應類

EventBloodActionOff #0 46

原應執行之傷害無效化

EventBloodActionChange #n1,n2 46

原應執行之傷害效果變更

(註：若為立即死亡者，只能以指定方式來改變所受之傷害。改變後之傷害執行視為直傷且無法觸發執行階段-46 事件。)

n1-(1.加成/2.減少/3.指定)

n2-(數值)

EventMoveActionOff #0 47

原應執行之距離變更無效化

EventHPLActionOff #0 48

原應執行之回復無效化

EventHPLActionChange #n1,n2 48

原應執行之回復效果變更

n1-(1.加成/2.減少/3.指定)

n2-(數值)

EventPersonResurrectActionOff #0 49

原應執行之人物角色復活效果無效化

EventActiveAIScore #n1,n2,[n3...] 主動技能 99

智能判斷型 AI 個別技能判斷評分

(註¹：自引數 n3 開始為選擇性填寫性質，目的為讓技能去挑選具貢獻之卡牌，並回報其編號。)

(註²：詳情請參閱第 9 章「主動技能之 AI 計算判斷設計」。)

n1-(1.推薦/2.無意見/3.不推薦)

n2-(1.一般型評分/2.殷數變化量優先型評分)

[n3...]- (個別推薦牌編號...)

II. 附加狀態

人物實際狀態類

PersonAddActualStatus #n1,n2,n3,n4

74

主動技能

被動技能

普通型

事件型

72/73/74/75

人物角色新增人物實際狀態

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(人物實際狀態之技能唯一識別碼)

n4-(效果回合數)

EventAddActualStatusData #n1,n2,n3,n4,n5,n6,n7,n8

人物實際狀態

普通型

事件型

72/73

人物實際狀態新增資料

(註：本指令為人物實際狀態於被新增時之擴充資料填入，原則上在執行階段-74 中使用，但該指令亦可在人物實際狀態週期內重複執行，每次執行指令均會立即變更目標人物之立繪資料。)

n1-(欲套用之大人物形象圖片路徑)

n2-(欲套用之小人物形象圖片路徑)

n3-(欲套用之小人物形象影子圖片路徑)

n4-(欲套用之顯示列人物頭像圖片路徑)

n5-(小人物形象影子座標 Left(使用者))

n6-(小人物形象影子座標 Top(使用者))

n7-(小人物形象影子座標 Left(電腦))

n8-(小人物形象影子座標 Top(電腦))

ActualStatusEnd #0

75

人物實際狀態

普通型

事件型

72/73/74/75

人物實際狀態宣告結束

(註¹：本指令執行後即宣告狀態終結，並非減少回合數。)

(註²：使用本指令會觸發執行階段-75 事件，並視為主動解除。)

PersonRemoveActualStatus #n1,n2

75

主動技能

被動技能

普通型

事件型

72/73/74/75

人物角色移除人物實際狀態

(註：使用本指令會觸發執行階段-75 事件，並視為被動解除。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

EventRemoveActualStatusActionOff #0

人物實際狀態

75

原應執行之人物實際狀態消除無效化

(註：若為主動解除時則無法觸發。)

異常狀態類

PersonAddBuff #n1,n2,n3,n4,n5

72

普通型

事件型

異常狀態

72/73

人物角色新增異常狀態

(註：若異常狀態已存在者則直接覆寫其效果設定。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(異常狀態之技能唯一識別碼)

n4-(效果變化量)

n5-(效果回合數)

BuffTurnEnd #0

73*

異常狀態

普通型

事件型

72/73

異常狀態宣告當回合結束

(註¹：本指令執行後會將狀態之回合數減少 1 回合，並非直接宣告結束。)

(註²：若使用本指令後回合數歸 0 者，會觸發執行階段-73 事件，並視為主動解除。)

PersonRemoveBuffAll #n1,n2

73

普通型

事件型

異常狀態

72/73

人物角色移除異常狀態(全部)

(註：使用本指令會觸發執行階段-73 事件，並視為被動解除。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

PersonRemoveBuffSelect #n1,n2,n3

73

普通型

事件型

異常狀態

72/73

人物角色移除異常狀態(指定)

(註：使用本指令會觸發執行階段-73 事件，並視為被動解除。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(異常狀態之技能唯一識別碼)

PersonBuffTurnChange #n1,n2,n3,n4,n5

73*

普通型

事件型

異常狀態

72/73

人物角色異常狀態變更回合數

(註：使用本指令若減少或指定回合數至 0 者，會觸發執行階段-73 事件，並視為主動解除。)

n1-(1.自身/2.對方)

n2-(第 x 位(VBEAtkingVSF 順序))

n3-(異常狀態之技能唯一識別碼)

n4-(1.增加/2.減少/3.指定)

n5-(數值)

EventRemoveBuffActionOff #0

異常狀態

73

原應執行之異常狀態消除無效化

(註：若為主動解除時則無法觸發。)

III. 其他事項

在腳本撰寫中，在指令名稱及指令引數上由「#」號分隔，引數與引數中間由「,」號分隔，如需傳回 2 個以上之指令者，指令與指令中間則以「=」號做分隔。

以下表格為插入事件卡效果指令「BattleInsertEventCard」所指定之事件卡標題字串一覽：

事件卡標題

單面事件卡							
劍 1	劍 2	劍 3	劍 4	劍 5	劍 6	劍 7	劍 8
槍 1	槍 2	槍 3	槍 4	槍 5	槍 6	槍 7	槍 8
防 1	防 2	防 3	防 4	防 5	防 7		
特 1	特 2	特 3	特 4	特 5			
移 1	移 2	移 3	移 4	移 5			
機會 1	機會 2	機會 3	機會 4	機會 5			
詛咒術 1	詛咒術 2	詛咒術 3	詛咒術 5				
HP 回復 1	HP 回復 2	HP 回復 3	聖水 ^{*1}				

雙面事件卡

劍 3/槍 1	劍 4/槍 2	劍 5/槍 3	劍 3/移 1	劍 4/移 1	劍 5/移 1	劍 3/防 1	劍 5/槍 5
防 3/移 1	防 4/移 1	防 5/移 1	特 1/防 1	特 2/防 2	特 3/防 3		
槍 3/劍 1	槍 4/劍 2	槍 5/劍 3	槍 3/移 1	槍 4/移 1	槍 5/移 1	槍 3/防 1	
移 1/特 1	移 2/特 2	移 3/特 3					

註^{*1}：使用聖水事件卡視同人物角色移除異常狀態(全部)，會觸發執行階段-73 事件，並視為聖水事件卡解除。

5. 腳本引數變數

為了能讓腳本進行遊戲場上之情勢做出判斷，遊戲系統提供了各式之場上資訊讓各技能體系腳本做為效果判斷之依據，以下開始為各引數變數之詳細說明：

腳本引數變數

變數名稱/內容	適用技能體系
VBEAtkingVSF(1 to 2,1 to 3,1 to 2) 雙方血量資訊一覽 (1.自身/2.對方, 1~3.場上順序排列-場上 1 位>場下待機 2 位, 1.當前血量/2.當前血量最大值)	全
VBEAtkingVSS(0 to 20) 戰鬥場上資訊一覽 0.目前角色之場上位置(VBEAtkingVSF 順序) 1.出牌數-自身 2.手牌數-自身 3.出牌數-對手 4.手牌數-對手 5.系統公骰擲骰後正骰數量(攻擊正面骰減去防禦正面骰) 6.當前距離(1.近距離/2.中距離/3.遠距離) 7.自身目前總骰數(系統公骰) 8.對手目前總骰數(系統公骰) 9.當前回合數 10.程式執行根目錄路徑 11.當前公牌數量 12.目前階段(1-DRAW/2-MOVE/3-ATK/4-DEF) 13.目前何方攻擊(1-自身/2-對方) 14.擲骰後自身正骰數量(系統公骰) 15.擲骰後對手正骰數量(系統公骰) 16.本回合何方先攻(0-未決定/1-自身/2-對方) 17.我方戰鬥總人數 18.敵方戰鬥總人數 19.我方目前卡牌最大數量上限 20.敵方目前卡牌最大數量上限	全

腳本引數變數

變數名稱/內容	適用技能體系
<p>AtkingPagetotVS(1 To 2, 1 To 5) 出牌種類及數值統計資料</p> <p>(1.自身/2.對方, 1.劍/2.防/3.移/4.特/5.槍)</p>	全
<p>VBEPersonVS(1 to 2,1 to 3,1 to 4,1 to 30,1 to 11) 人物角色總資訊一覽</p> <p>(1.自身/2.對方, 1~3.第 x 位人物(VBEAtkingVSF 順序), 1.基本資料系/2.座標系/3.技能系/4.對話系, 1~30.各類別(參照下方說明), 1~11.各項目(參照下方說明))</p> <p>基本資料系</p> <p>1.名稱類(1.中文名稱/2.英文名稱/3.稱號) 2.牌等級類(1.等級類別/2.等級數) 3.人物能力值類(1.HP/2.ATK/3.DEF/4.人物事件卡顏色格數字串/5.能力資訊是否為新樣式[1.是/0.否]) 4.ID 類(1.VBEID/2.ULID/3.AI 版本號) 5.人物圖片路徑類(1.小人物圖路徑/2.小人物影子圖路徑/3.大人物圖路徑/4.小人物頭像圖路徑/5.人物卡片圖路徑)</p> <p>座標系</p> <p>1.小人物立繪類(1~4.(無資料)/5.影子 Left/6.影子 Top) 2.大人物立繪類(1.Height/2.Width/3.Top/4.移動最大 Left/5.是否直接移動到底[1.是/0.否]) 3.技能字體大小類(1~4.第 n 個個別技能說明字體大小/5.是否套用預設值 13[1.是/0.否])</p> <p>技能系</p> <p>1~4.第 1~4 種主動技能詳細類(1.名稱/2.階段/3.距離/4.卡片/5.效果/6.卡片字型大小(選)/7.效果字型大小(選)/8.階段代碼/9.距離代碼/10.卡片代碼/11.技能唯一識別碼) 5~8.第 1~4 種被動技能詳細類(1.名稱/2.效果/3.技能唯一識別碼)</p> <p>對話系</p> <p>1~20.第 n 個相對角色對話類(1.對話內容/2.相對角色名) 21~30.第 n 個無相對角色對話類(1.對話內容)</p>	<p>主動技能 被動技能 人物實際狀態</p>

腳本引數變數

變數名稱/內容	適用技能體系
VBEPageCardNumVS(1 To ?, 1 To 6) 戰鬥系統場上卡牌資訊一覽 (註：對於場上之卡牌張數為不確定數，建議在迴圈搜尋時採用「UBound(VBEPageCardNumVS, 1)」之方法。) <div style="text-align: right;">(第 x 編號,</div> 1.正面類型/2.正面數值/3.反面類型/4.反面數值/5.(1)自身-(2)對方/ 6.(1)手牌-(2)出牌-(3)墓地牌-(4)牌堆-(5)未使用 (註：卡牌類型如下： 1.劍/2.防/3.移/4.特/5.槍/6.機會/7.詛咒術/8.回復/9.聖水/0.其他) AtkingckVSS(1 to 8,1 to 3) 主動技能/被動技能資訊一覽(一般) (1~4 人物自身主動技能項目;5~8 人物自身被動技能項目, 1.技能啟動標記/2.這回合間啟動次數/3.這場戰鬥間啟動次數)	主動技能 被動技能 人物實際狀態
AtkingckVSF(1 to 8,1 to 1) 主動技能/被動技能資訊一覽(字串) (1~4 人物自身主動技能項目;5~8 人物自身被動技能項目, 1.技能備註資訊)	主動技能 被動技能
VBEPersonBuffVSF(1 to 2,1 to 3, 1 To ?, 1 To 2) 異常狀態資訊一覽(一般) (註：對於場上人物之異常狀態數量為不確定數，建議在迴圈搜尋時採用「UBound(VBEPersonBuffVSF, 3)」之方法。) <div style="text-align: right;">(1.自身/2.對方,</div> <div style="text-align: right;">第 x 位(VBEAtkingVSF 順序),</div> <div style="text-align: right;">第 x 個異常狀態,</div> 1.狀態數值/2.狀態統計數(剩餘回合/累計))	主動技能 被動技能 人物實際狀態
VBEPersonBuffVSS(1 to 2,1 to 3, 1 To ?) 異常狀態資訊一覽(技能唯一識別碼) (註：對於場上人物之異常狀態數量為不確定數，建議在迴圈搜尋時採用「UBound(VBEPersonBuffVSS, 3)」之方法。) <div style="text-align: right;">(1.自身/2.對方,</div> <div style="text-align: right;">第 x 位(VBEAtkingVSF 順序),</div> <div style="text-align: right;">第 x 個異常狀態-技能唯一識別碼) </div>	主動技能 被動技能 人物實際狀態

腳本引數變數

變數名稱/內容	適用技能體系
VBEActualStatusVS(1 to 2,1 to 3,1 to 2) 人物實際狀態資訊一覽 (1.自身/2.對方, 第 x 位(VBEAtkingVSF 順序), 1.技能唯一識別碼/2.人物實際狀態名稱)	主動技能 被動技能
BuffPersonType 該角色目前場上狀態 (1)場上角色 (2)待機角色	異常狀態
ActualStatusPersonType 該角色目前場上狀態 (1)場上角色 (2)待機角色	人物實際狀態
PassivePersonType 該角色目前場上狀態 (1)場上角色 (2)待機角色	被動技能
VBEVSBuffNum(1 to 2) 異常狀態自身資訊一覽 1.異常狀態自身變化量數 2.異常狀態自身回合數	異常狀態
VBEVSStageNum(?) 執行階段多用途紀錄資訊 (註：對於多用途紀錄資訊變數為不確定數，且僅部分之執行階段才會提供資訊，詳見以下之列表：) 執行階段-41(人物交換) (1)執行效果方(1.自身/2.對方) (2)交換前人物編號(VBEAtkingVSF 順序) (3)交換後人物編號(VBEAtkingVSF 順序)	全

執行階段-46(傷害)

- (1)受到傷害方(1.自身/2.對方)
- (2)受到傷害方人物編號(VBEAtkingVSF 順序)
- (3)受到傷害之形式(1.毆傷/2.直傷/3.立即死亡)
- (4)受到傷害之數值
- (5)觸發事件方(0.系統/1.自身/2.對方)
- (6)觸發狀態類別(0.系統/1.主動技能/2.被動技能/3.異常狀態/4.人物實際狀態)

執行階段-47(移動)

- (1)變更前距離
- (2)變更後距離

執行階段-48(回復)

- (1)回復方(1.自身/2.對方)
- (2)回復方人物編號(VBEAtkingVSF 順序)
- (3)回復之數值
- (4)觸發事件方(0.系統/1.自身/2.對方)
- (5)觸發狀態類別(0.系統/1.主動技能/2.被動技能/3.異常狀態/4.人物實際狀態)

執行階段-2/3/4/70/71(普通型-移動前)

- (1)自身目前總計之移動數值
- (2)對方目前總計之移動數值
- (3)自身目前移動階段選擇項目(0:無/1:向遠/2:回復/3:向近/4:更換角色)
- (4)對方目前移動階段選擇項目(0:無/1:向遠/2:回復/3:向近/4:更換角色)

執行階段-62(技能效果進行多次擲骰時)

- (1)自身總骰數
- (2)對手總骰數
- (3)擲骰後自身正骰數量
- (4)擲骰後對手正骰數量
- (5)擲骰後正骰數量(攻擊正面骰減去防禦正面骰)

執行階段-73(異常狀態消滅時)

- (1)是否為被動消除(0.否/1.是/2.聖水事件卡)

腳本引數變數

變數名稱/內容	適用技能體系
執行階段-74(人物實際狀態-新增) (1)原引發之技能所要求之效果回合數	
執行階段-75(人物實際狀態消滅時) (1)是否為被動消除(0.否/1.是)	
執行階段-76/77(人物角色附加狀態增加/解除時) (1)觸發事件方(1.自身/2.對方) (2)觸發狀態類別(1.異常狀態/2.人物實際狀態) (3)該觸發狀態之技能唯一識別碼(Type:String)(若為異常狀態全體解除者，則為空白)	
執行階段-42/43/44(特殊型-各階段出牌時) (1)觸發事件方(1.自身/2.對方)	

6. 腳本體系類型

在 UnlightVBE-QS 遊戲系統中，會影響人物角色之技能體系主要為以下 4

個：

- I. 主動技能：人物角色所擁有，也是最基礎之技能體系，需要透過指定之出牌條件來發動，每個人物角色最多擁有 4 個主動技能。
- II. 被動技能：人物角色所擁有，相對於主動技能，被動技能擁有在不特定之時間點下發動，且無須設定出牌條件，每個人物角色最多擁有 4 個被動技能。
- III. 異常狀態：獨立之技能體系，會附著在人物角色身上，需要透過其他技能體系來引導發動。
- IV. 人物實際狀態：獨立之技能體系，相對於異常狀態之附著特性，人物實際狀態為在特定時間內對人物角色造成其實體改變，需要透過其他技能體系來引導發動，每個人物角色最多擁有 1 個人物實際狀態。

為了能讓各位更了解這些技能體系之歸屬，以下表格範例為引用 Unlight

官方人物角色-復活角色艾茵 N1 (ULID:2111) 之各技能歸屬：

艾茵(N1)

主動技能	一顆心、七個願望、十三隻眼
被動技能	兩個身體
人物實際狀態	貓化狀態

在遊戲設計中，人物實際狀態為最具有權力及影響力之技能體系，其次是異常狀態、被動技能，最後是主動技能。在每個執行階段中，人物角色會依照技能體系權力大小之順序來執行效果：「**人物實際狀態>異常狀態>被動技能>主動技能**」。

另外在攻擊/防禦階段之**骰數變化量**計算之執行階段-45事件中，各個技能體系所分別給予之總骰數變化量計算則依照影響力之順序來**先後**計算加成：「**主動技能>被動技能>異常狀態>人物實際狀態**」，另因執行階段-45是系統計算雙方之總骰數變化量的唯一事件，如果在雙方都有讓對方具總骰數變化量的影響時，人物角色計算加成之順序如下：「**自身之主動技能>自身之被動技能>自身之異常狀態>自身之人物實際狀態>對手之主動技能>對手之被動技能>對手之異常狀態>對手之人物實際狀態**」。

以下表格範例為引用 Unlight 官方人物角色-史塔夏 LV5 (ULID:335 , VBEID:N00505) 之實際計算狀況，假設史塔夏已變身殺戮狀態，對手全員活著，在攻擊階段於近距離出一張劍 6 行動卡的情況下，自身之總骰數計算如下：

史塔夏(LV5)自身總骰數變化量加成

攻擊白值	行動卡	主動技能	被動技能	異常狀態	人物實際狀態	總骰數
4 (LV5)	+6 (劍 6)	+12 (愚者之手)			*2 (殺戮狀態)	44

7. 腳本及卡片人物資訊文件格式

I. 卡片人物資訊 (UnlightVBE Card Information)

卡片人物資訊文件 (*.uleci) 為記載人物角色之各式基本資訊之純文

字文件，儲存在 character 資料夾中以人物角色名稱作為個別人物角

色之資料夾下。以下表格為引用 UnlightVBE 中自創人物角色-蕾

(VBEID:S001) 之卡片人物資訊文件基礎架構：

卡片人物資訊文件架構

架構程式碼	說明
Q4VX435S MenuName= 蕾 MenuList=LV5=R1=R2=R3=R4=R5 EndFirst	初始架構，紀錄選單及文件內所包含的等級
StartPerson=LV5=蕾 cardjpg= Ria\RiaLV5.jpg personhp=9 personatk=5 persondef=10 EndPerson	主資訊架構，紀錄角色某一等級的基本資訊
StartTalk=蕾=LV=R EndTalk	對話架構，紀錄角色的人物對話資訊

在一份文件中必須擁有初始架構及至少 1 項等級之主資訊架構，對話

架構則為選擇性填寫。

以下表格為架構中每項設定之詳細說明：

卡片人物資訊文件架構設定項說明（初始架構）

標題	內容	說明
Q4VX435S		VBE-QS 1.0 版本文件檢查碼 (G15d466E : VBE -1.0 版本文件檢查碼 XzPF5922 : VBE -2.0 版本文件檢查碼)
MenuName	蕾	人物名稱
MenuList	LV5=R1=R2=R3=R4=R5	此文件所納入的角色等級。 (等級與等級中間以「=」隔開)
EndFirst		結束初始架構文字

卡片人物資訊文件架構設定項說明（主資訊架構）

標題	內容	說明
StartPerson	LV5=蕾	開始架構文字 LV5：本段落資訊所指的角色等級 蕾：人物名稱
cardjpg	Ria\RiaLV5.jpg	卡面圖片（註 ³ ）
personhp	9	血量（白值）
personatk	5	攻擊力（白值）
persondef	10	防禦力（白值）
cardInfisNewType	0	角色能力資訊是否為新樣式 [0-否/1-是] (新舊樣式會影響數字之位置顯示)
personname	蕾	角色名稱
personengname	Ria	角色英文名稱
personpname	Aria	角色稱號
personlevel1	LV	等級分類 (例：LV、R)
personlevel2	5	等級數字
cardid	S00105	VBE 角色 ID（註 ¹ ）
persontg	357000	角色卡色格（註 ² ）
personbig	Ria\Riaperson1.png	大人物立繪（註 ³ ）
personmini	Ria\Riamini1.png	小人物立繪（註 ³ ）
personf	Ria\Riaf1.png	顯示列臉圖（註 ³ ）
personsmalldown	Ria\Riaminidown1.png	小人物立繪影子（註 ³ ）

卡片人物資訊文件架構設定項說明 (主資訊架構)

標題	內容	說明
atkingfontck	0	角色技能欄字體大小是否為預設值 [1-預設值大小(12)/0-自訂] (自訂時須加上以下 4 列)
atkingfont1	10	第 1 個主動技能字體大小 (7~)
atkingfont2	10	第 2 個主動技能字體大小 (7~)
atkingfont3	10	第 3 個主動技能字體大小 (7~)
atkingfont4	10	第 4 個主動技能字體大小 (7~)
atkingjpgleftallzero	0	是否由系統自動控制大人物立繪移動位置 (註 ⁴) [0-自動/1-對齊邊框]
smalldownleftus	0	使用者方小人物立繪影子座標 (Left)
smalldowntopus	-30	使用者方小人物立繪影子座標 (Top)
smalldownleftcom	40	電腦方小人物立繪影子座標 (Left)
smalldowntopcom	-30	電腦方小人物立繪影子座標 (Top)
atkingname(1)	PNAKS00101=atking1.ulevsa	第 1~4 項人物技能腳本連結 (註 ⁵)
...
atkingname(5)	PNPES00101=passive1.ulevsp	第 5~8 項人物技能腳本連結 (註 ⁵)
...

卡片人物資訊文件架構設定項說明（主資訊架構）

標題	內容	說明
atkingfont(1)	8	第 1~4 項主動技能發動條件 字體大小 (像是有「無合計 1 以上、無合計 1 以上、無合計 1 以上」超過 2 個條件者適用，建議大小為 8)
atkingdfont(1)	8	第 1~4 項主動技能發動條件 字體大小 (效果文字太長以致系統無法顯示全部者適用，建議大小為 8)
EndPerson		結束架構文字

(若該選項有紅框者，則為選擇性填寫。)

卡片人物資訊文件架構設定項說明（對話架構）

標題	內容	說明
StartTalk	蕾=LV=R	開始架構文字 蕾：人物名稱 LV、R：該對話架構所適用之角色等級類別 (若要含 2 個以上等級類別，中間以「=」相隔)
TalkA1	為什麼要那麼傻呢，笨蛋。=凱爾	第 n 個相對角色類型對話(至多 20 個) 凱爾-本對話所指定之人物 *附註：對話若有 2 個以上指定人物者，人物與人物中間以「&」相隔。 ex.艾伯李斯特對阿貝爾及利恩的對話「時間是會流逝的，不管在哪裡。」 則其代碼為「阿貝爾&利恩」
...	...	
TalkB1	吶，聽我唱首歌吧！	第 n 個無相對角色類型對話(至多 10 個)
...	...	
EndTalk		結束架構文字

(註¹)：VBE 角色 ID 為人物角色之唯一編號字串，本人制定時共為 6 碼，開發者亦可自行使用自行編製之編號字串，但盡量具唯一性。

以下表格為 VBE 角色 ID 之編號說明：

VBE 角色 ID 編號說明

項目		說明
主分類	N	官方人物角色 (不限官方分類)
	S	自創人物角色
人物主編號	001	編號自 001~999
分項目號	01	編號自 01~99 通常會對應人物角色等級數字

(註²)：角色卡色格字串共 6 碼，對應人物角色之 6 格事件卡格：

人物角色卡色格

代號	對應顏色
0	(無)
1	紅
2	綠
3	藍
4	紫
5	白
6	黑
7	黃

當遇到人物角色非 6 格事件卡格時，先將擁有之事件卡格編號，剩下之長度則補 0。

(註³)：所有圖片都要放在「gif\<人物角色名>」之資料夾內，且所有之圖片路徑需加入母資料夾名稱 (人物角色名)。

(註⁴)：部分人物角色立繪在自動控制時會有顯示上的不健全，主要原因為該立繪圖片主體有削邊之情形發生。

(註⁵)：技能腳本連結共有 2 個分項目，第 1 個為該技能之技能唯一識別碼，第 2 個為該技能腳本之檔名 (含副檔名)，第 1~4 項人物技能為主動技能，第 5~8 項人物技能則為被動技能。

II. 技能腳本文件

對於 4 個技能體系腳本文件架構而言，其實大部分都是相同的，只是在 Function 的名稱，以及引數變數上面會有所差異。

以下表格為技能體系腳本文件之分類：

腳本文件副檔名分類

技能體系	文件副檔名
主動技能 (Active)	*.ulevsa
被動技能 (Passive)	*.ulevsp
異常狀態 (Buff)	*.ulevsbf
人物實際狀態 (Character Actual Status)	*.ulevsc

腳本文件存放路徑分類

技能體系	文件存放路徑
主動技能 (Active)	\Character\<人物角色名>
被動技能 (Passive)	\Character\<人物角色名>
異常狀態 (Buff)	\Buff\<異常狀態名>
人物實際狀態 (Character Actual Status)	\CharacterActualStatus\<人物實際狀態名>

注意以上之路徑都是以主程式所在資料夾作為開始之路徑。

以下表格為引用 UnlightVBE 中自創人物角色-蕾 (VBEID:S001) 之主動技能腳本文件基礎架構：

技能腳本文件架構

架構程式碼範例	說明
<pre> Function main(a) Select Case a Case 1 main = "PNAKS00101" Case 2 main = "42#45#11" Case 3 main = "輪旋曲-琉璃色的微風" Case 4 main = "攻擊#1" Case 5 main = "近距離#001" Case 6 main = "劍合計 4 以上#14a" Case 7 main = "ATK+ 6 · 對手這回合有效防禦值-3" End Select End Function Function atking(a, VBEPersonVS(), VBEPageCardNumVS(), AtkingPagetotVS(), VBEPersonBuffVSF(), VBEPersonBuffVSS(), AtkingckVSS(), AtkingckVSF(), VBEAtkingVSF(), VBEAtkingVSS(), VBEActualStatusVS(), VBEVSSStageNum()) Select Case a Case 42 If VBEAtkingVSS(6) = 1 Then If AtkingPagetotVS(1, 1) >= 4 Then If AtkingckVSS(1, 1) = 0 Then atking = atking & "AtkingLineLight#1" & "=" atking = atking & "AtkingTurnOnOff#1" & "=" End If Else If AtkingckVSS(1, 1) = 1 Then atking = atking & "AtkingLineLight#2" & "=" atking = atking & "AtkingTurnOnOff#2" & "=" End If End If End If Case 45 atking = atking & "EventTotalDiceChange#1,1,6" & "=" atking = atking & "EventTotalDiceChange#2,2,3" & "=" Case 11 atking = atking & "AtkingStartPlay#" & "\gif\Ria\atking1.png,0" & "=" atking = atking & "AtkingLineLight#2" & "=" atking = atking & "AtkingTurnOnOff#2" & "=" End Select End Function </pre>	<p>初始架構，紀錄技能之基本資訊</p> <p>主架構，紀錄技能之效果撰寫</p>

注意不管是哪種技能技系，在一份文件中必須要擁有以上之 2 個架構。

在不同之技能體系中，初始架構之 Function 名稱都為 main，不過在主架構 Function 之名稱會有所差異：

主架構 Function 名稱分類

技能體系	Function 名稱
主動技能 (Active)	atking
被動技能 (Passive)	passive
異常狀態 (Buff)	buff
人物實際狀態 (Character Actual Status)	ActualStatus

初始架構之意義為一個技能之標頭資訊，以下就 4 個技能體系之不同的標頭資訊做說明：

初始架構標頭資訊說明 (主動技能)

(引用:自創人物角色-蕾 LV5 (VBEID:S00105) 主動技能-輪旋曲-琉璃色的微風)

編號	內容	說明
1	PNAKS00101	技能唯一識別碼 (註 ¹)
2	42#45#11	欲使用之執行階段號 (註 ²)
3	輪旋曲-琉璃色的微風	主動技能名稱
4	攻擊#1	主動技能發動階段 1-攻擊階段 2-防禦階段 3-移動階段
5	近距離#001	主動技能發動距離 0-該技能是否適用於遠距離(0-否/1-是) 0-該技能是否適用於中距離(0-否/1-是) 1-該技能是否適用於近距離(0-否/1-是)
6	劍合計 4 以上#14a	主動技能發動條件 (註 ³) 1-條件發動牌類型 (1-劍/2-盾/3-移/4-特/5-槍/6-其他) 4-條件發動牌數值 (0~9/a->10/b->?) a-條件發動數值之採取方式 (a-合計以上/b-指定等於/c-合計以下)
7	ATK+ 6 · 對手這回合有效防禦值-3	主動技能發動效果
8	(無)	主動技能於角色待命狀態時仍可執行之啟動旗標 (ON-開啟/非 ON 者則關閉)

(若該選項有紅框者，則為選擇性填寫。)

初始架構標頭資訊說明 (被動技能)

(引用:官方人物角色-莉莉 LV1 (ULID:40001 、 VBEID:N02701) 被動技能-小公主)

編號	內容	說明
1	PNPEN02701	技能唯一識別碼 (註 ¹)
2	1#22#55	欲使用之執行階段號 (註 ²)
3	小公主	被動技能名稱
4	20%機率受到的傷害為 0 。只會在 HP5 以上時發動。	被動技能發動效果

初始架構標頭資訊說明 (異常狀態)

(引用:增益系異常狀態-ATK+)

編號	內容	說明
1	BUFFN00101	技能唯一識別碼 (註 ¹)
2	45#53	欲使用之執行階段號 (註 ²)
3	ATK+	異常狀態名稱
4	\Buff\atkup\atkupnew.gif	異常狀態圖片路徑 (註 ⁴)

初始架構標頭資訊說明 (人物實際狀態)

(引用:官方人物角色-史塔夏 LV5 (ULID:335 、 VBEID:N00505) 人物實際狀態-殺戮模式)

編號	內容	說明
1	UCASN00101	技能唯一識別碼 (註 ¹)
2	74#45#54	欲使用之執行階段號 (註 ²)
3	殺戮狀態	人物實際狀態名稱

(註¹)：技能唯一識別碼為每個技能之唯一編號字串，本人制定時共為 10 碼，開發者亦可自行使用自行編製之編號字串，但需要具唯一性，否則在遊戲執行時會造成錯誤。以下表格技能唯一識別碼之編號說明：

技能唯一識別碼編號說明

項目		說明
主分類	PNAK	主動技能代號
	PNPE	被動技能代號
	BUFF	異常狀態代號
	UCAS	人物實際狀態代號
技能從屬編號	N001	主動 / 被動技能：欲使用該技能之人物角色編號 異常狀態 / 人物實際狀態：個別效果體系編號 (001~999)
分項目號	01	編號自 01~99

(註²)：對於執行階段點的使用採登記制，如該技能需要使用到特定之執行階段點，需要在本欄設定上註記執行階段號，如有 2 個以上者，中間以「#」號間隔。

(註³)：若該技能有 2 個發動條件以上者，代碼與代碼中間以「&」相隔，至多 5 個。EX：發動條件為「槍 1、槍 2、槍 3」時，則代碼為「51b&52b&53b」。

(註⁴)：異常狀態之圖片路徑是以主程式所在資料夾作為開始之參照。

主架構包含程式之引數變數，登記之執行階段號以及效果程式碼之部分，而在主動技能方面則還包括 AI 判斷設計之內容。本章節主要討論主架構之大綱部分，有關於其他之各部分會在其他之章節再做說明，以下表格開始為主架構大綱部分及效果程式碼之說明：

(因版面狹小，部分程式碼會採取換行之寫法)

主架構效果程式碼說明 (主動技能)	
(引用:自創人物角色-蕾 LV5 (VBEID:S00105) 主動技能-輪旋曲-琉璃色的微風)	
架構程式碼	
<pre> Function atking(a, VBEPersonVS(), VBEPageCardNumVS(), AtkingPagetotVS(), VBEPersonBuffVSF(), VBEPersonBuffVSS(), AtkingckVSS(), AtkingckVSF(), VBEAtkingVSF(), VBEAtkingVSS(), VBEActualStatusVS(), VBEVSSStageNum()) Select Case a Case 42 '攻擊階段(出牌時) If VBEAtkingVSS(6) = 1 Then '當現在為近距離時 If AtkingPagetotVS(1, 1) >= 4 Then '如果出的劍卡總數值超過 4 時 If AtkingckVSS(1, 1) = 0 Then '當技能未啟動時 atking = atking & "AtkingLineLight#1" & "=" atking = atking & "AtkingTurnOnOff#1" & "=" End If Else If AtkingckVSS(1, 1) = 1 Then '當出牌條件未符合且技能已啟動時 atking = atking & "AtkingLineLight#2" & "=" atking = atking & "AtkingTurnOnOff#2" & "=" End If End If End If Case 45 '總骰數變化量影響 atking = atking & "EventTotalDiceChange#1,1,6" & "=" atking = atking & "EventTotalDiceChange#2,2,3" & "=" Case 11 '發動技能動畫 atking = atking & "AtkingStartPlay#" & "\gif\Ria\atking1.png,0" & "=" atking = atking & "AtkingLineLight#2" & "=" atking = atking & "AtkingTurnOnOff#2" & "=" End Select End Function </pre>	

主架構效果程式碼說明 (被動技能)

(引用:官方人物角色-莉莉 LV1 (ULID:40001 、 VBEID:N02701) 被動技能-小公主)

架構程式碼

```
Function passive(a, VBEPersonVS(), VBEPageCardNumVS(), AtkingPagetotVS(),  
VBEPersonBuffVSF(), VBEPersonBuffVSS(), AtkingckVSS(), AtkingckVSF(),  
VBEAtkingVSF(), VBEAtkingVSS(), VBEActualStatusVS(), PassivePersonType,  
VBEVSSStageNum())  
    Select Case a  
        Case 0 '發牌階段  
            If VBEAtkingVSF(1, 1, 1) >= 5 And PassivePersonType = 1 Then  
                '如果自身血量大於 5 且為場上角色時  
                passive = passive & "AtkingLineLight#1" & "="  
                passive = passive & "AtkingTurnOnOff#1" & "="  
            End If  
        Case 22 '攻擊/防禦階段最後執行點-其他骰數有關類型  
            If PassivePersonType = 1 And VBEAtkingVSS(12) = 4 Then  
                '如果現在為場上角色且為防禦階段時  
                Dim m  
                Randomize Timer  
                m = Int(Rnd() * 100) + 1 '機率取樣  
                If (AtkingckVSS(1, 1) = 1 And m >= 53 And m <= 92) Or _  
                    (AtkingckVSS(1, 1) = 0 And m Mod 5 = 0) Then '=40%/20%之機率  
                    passive = passive & "AtkingTrueDiceControl#3,0" & "="  
                    passive = passive & _  
                        "BattleSendMessage#因為小公主的效果・傷害被無效了。" & "="  
                End If  
            End If  
        Case 55 '回合結束階段  
            If PassivePersonType = 1 Then '如果現在為場上角色時  
                passive = passive & "AtkingLineLight#2" & "="  
                passive = passive & "AtkingTurnOnOff#2" & "="  
            End If  
    End Select  
End Function
```

主架構效果程式碼說明 (異常狀態)

(引用:增益系異常狀態-ATK+)

架構程式碼

```
Function buff(a, AtkingPagetotVS(), VBEAtkingVSF(), VBEAtkingVSS(),  
VBEVSBuffNum(), BuffPersonType, VBEVSSStageNum())  
  
    Select Case a  
  
        Case 45 '總骰數變化量影響  
  
            If BuffPersonType = 1 Then '如果附身之人物角色在場上的話  
  
                If VBEAtkingVSS(12) = 3 Then '如果現在為攻擊階段的話  
  
                    If (AtkingPagetotVS(1, 1) > 0 And VBEAtkingVSS(6) = 1) Or _  
                        (AtkingPagetotVS(1, 5) > 0 And VBEAtkingVSS(6) > 1) Then  
  
                        '如果出的劍卡總數值大於 0 且為近距離 ,  
  
                        '或者出的槍卡總數值大於 0 且為中、遠距離時  
  
                        buff = buff & "EventTotalDiceChange#1,1," & _  
                            VBEVSBuffNum(1) & "="  
  
                    End If  
  
                End If  
  
            End If  
  
        Case 53 '回合結束階段  
  
            If BuffPersonType = 1 Then '如果附身之人物角色在場上的話  
  
                buff = buff & "BuffTurnEnd#0" & "="  
  
            End If  
  
        End Select  
  
    End Function
```

主架構效果程式碼說明 (人物實際狀態)

(引用:官方人物角色-史塔夏 LV5 (ULID:335 、 VBEID:N00505) 人物實際狀態-殺戮模式)

架構程式碼

```
Function ActualStatus(a, VBEPersonVS(), VBEPageCardNumVS(),  
AtkingPagetotVS(), VBEPersonBuffVSF(), VBEPersonBuffVSS(), VBEAtkingVSF(),  
VBEAtkingVSS(), ActualStatusPersonType, VBEVSStageNum())  
    Select Case a  
        Case 74 '人物實際狀態-新增事件  
            '新增殺戮狀態立繪資料  
            ActualStatus = ActualStatus & "EventAddActualStatusData#" & _  
                "\CharacterActualStatus\KillerMode\personb.png," & _  
                "\CharacterActualStatus\KillerMode\personmini.png," & _  
                "\CharacterActualStatus\KillerMode\personminidown.png," & _  
                "\CharacterActualStatus\KillerMode\personf.png," & _  
                "-90,-60,90,-60" & "="  
        Case 45 '總骰數變化量影響  
            If ActualStatusPersonTypeVSS = 1 Then '如果現在為場上角色時  
                ActualStatus = ActualStatus & "EventTotalDiceChange#1,3,2" & "="  
            End If  
        Case 54 '回合結束階段  
            If ActualStatusPersonTypeVSS = 1 Then '如果現在為場上角色時  
                ActualStatus = ActualStatus & "ActualStatusEnd#0" & "="  
            End If  
    End Select  
End Function
```

8. 主動技能之 AI 計算判斷設計

自 UnlightVBE Version 7 開始實裝之 AI 系統-智能判斷型人工智慧 AI-「夢」，以透過手牌之排列組合計算作為判斷出牌之依據，本章為說明 AI 系統之個別技能計算部分。

在 UnlightVBE-QS 中，AI 在計算卡牌的排列組合期望值之時，為了能更加貼切每個人物角色之需求，會徵求人物角色之各主動技能之意見，而各技能則依據各自之判斷，回傳其判斷結果給 AI 系統。

在實際運作之中，與技能之間之意見傳遞會透過執行階段系統-99 號來執行，與效果程式碼不同的是，在執行階段-99 之腳本引數有關卡牌變數部分，會直接套用當前該排列組合之計算資訊，而非實際在場上之情況，主動技能僅須直接就當前之資訊給予判斷即可。

以下為在執行階段-99 之中，技能所要回傳之指定指令內容：

執行階段-99 指定指令			
EventActiveAIScore	#n1,n2,[n3...]	主動技能	99
智能判斷型 AI 個別技能判斷評分			
(註：自引數 n3 開始為選擇性填寫性質，目的為讓技能去挑選具貢獻之卡牌，並回報其編號。)			
n1-(1.推薦/2.無意見/3.不推薦)			
n2-(1.一般型評分/2.殷數變化量優先型評分)			
[n3...]- (個別推薦牌編號...)			

對於一個卡牌排列組合，如果有任何主動技能「推薦」的話，將讓該組合的期望值獲得提高，而 AI 系統最終將會挑選出最高期望值的排列組合，當作出牌的依據。反之若有任何主動技能回報「不推薦」的話，將使該組合期望值評比降低，進而淘汰這些不適合的排列組合。

若主動技能並無納入 AI 計算判斷設計，或者是該組合被回報為「無意見」的話，AI 系統將不會對這些排列組合作任何期望值上的技能加成，但技能仍可以回報具有貢獻的卡牌編號，這依然會有因 AI 系統的篩選而被選中的可能性存在。

在判斷評分類型上分成一般型評分與骰數變化量優先型評分，主要不同在於期望值的加權方式，一般型評分只會做期望值固定加成，對於以骰量加成為主要效果之技能來說，一般型評分的加成結果是相同的，這時就可選擇後者之評分方式，因骰數變化量優先型評分著重於總骰數變化量之加成取向，AI 系統會參考該技能於執行階段-45 之骰量加成數據，來當作期望值加成的依據。

從另一個角度來說，個別技能判斷評分講求的是個人在出牌方面的經驗，這並沒有所謂的標準寫法，而一個好的判斷評分會讓玩家在對戰上面會有更佳的使用者體驗。以下為了能讓各位更加了解個別技能判斷評分的運作，本人將示範設計 Unlight 官方人物角色-雪莉 LV5 (ULID:105，VBEID:N00105) 之主動技能「自殺傾向」作為這次的範例。


我們先來看一下該技能的基本資料：

雪莉 LV5-主動技能「自殺傾向」


項目	內容
技能唯一識別碼	PNAKN00101
名稱	自殺傾向
階段	攻擊
距離	全距離
發動條件	特合計 1 以上
發動效果	HP-特，ATK+特 x5

自殺傾向是一個能夠創造高攻擊力的一項技能，雖然要以自身割血作為代價，但是當能夠一次累積相當數量的特卡之時，其對對手之破壞力就不在話下了。

接下來我們加入一些戰鬥的條件假設如圖(8-1)所示，雪莉在第 3 回合近距離下先攻，在剩餘 8 滴血的情況下，要對圖片中之手牌作出牌的判斷選擇。如果以技能最單純的目的，也就是創造高攻擊力去想的話，最理想的情況就是把所有的特卡都出出去，如圖(8-2)所示，這種判斷方法是否可行呢？




圖(8-1)



圖(8-2)

答案我想各位都很清楚，這種出牌可謂真正的「自殺」，根本毫不可行，但 Unlight 官方曾經有一段時間的 PVP-AI 真的是這種判斷方式。

要避免這樣的「自殺」攻擊，我們需要在針對特卡的數量來下更多判斷，當像圖(8-2)這種已經會危害到人物角色生命的這種出牌排列組合，我們應當對 AI 系統作出「不推薦」的一個判斷評分，反之在適當的特卡數量之出牌排列組合下，我們就能夠向 AI 系統進行「推薦」了。圖(8-3)是本人覺得最適當的出牌排列組合。



圖(8-3)

綜合以上的思考，最後我們要將這種判斷實行在腳本之中，以下為原效果程式碼加上 AI 判斷設計程式碼之主架構說明：

主架構-效果&AI 判斷設計程式碼說明 (主動技能)

(引用:官方人物角色-雪莉 LV5 (ULID:105 · VBEID:N00105) 主動技能-自殺傾向)


架構程式碼

```
Function atking(a, VBEPersonVS(), VBEPageCardNumVS(), AtkingPagetotVS(),  
VBEPersonBuffVSF(), VBEPersonBuffVSS(), AtkingckVSS(), AtkingckVSF(), VBEAtkingVSF(),  
VBEAtkingVSS(), VBEActualStatusVS(), VBEVSSStageNum())  
Select Case a  
Case 42 '攻擊階段(出牌時)  
If AtkingPagetotVS(1, 4) >= 1 Then '如果出的特卡總數值超過 1 時  
If AtkingckVSS(1, 1) = 0 Then '當技能未啟動時  
atking = atking & "AtkingLineLight#1" & "="  
atking = atking & "AtkingTurnOnOff#1" & "="  
End If  
Else  
If AtkingckVSS(1, 1) = 1 Then '當出牌條件未符合且技能已啟動時  
atking = atking & "AtkingLineLight#2" & "="  
atking = atking & "AtkingTurnOnOff#2" & "="  
End If  
End If  
Case 45 '總數變化量影響  
atking = atking & "EventTotalDiceChange#1,1," & (AtkingPagetotVS(1, 4) * 5) & "="  
'增加出牌之特總數*5 之總數量  
Case 11 '發動技能動畫  
atking = atking & "AtkingStartPlay#" & "\gif\Sheri\atking1.jpg,0" & "="  
atking = atking & "AtkingLineLight#2" & "="  
atking = atking & "AtkingTurnOnOff#2" & "="  
Case 61 '技能動畫中效果實施  
atking = atking & "PersonBloodControl#1,1,1," & AtkingPagetotVS(1, 4) & "="  
'對自己造成出牌之特總數的傷害  
Case 99 'AI 判斷設計  
If VBEAtkingVSS(12) = 3 And VBEAtkingVSS(13) = 1 Then  
'如果現在為攻擊階段/我方攻擊時  
If AtkingPagetotVS(1, 4) >= 1 And _  
VBEAtkingVSF(1, 1, 1) - AtkingPagetotVS(1, 4) > 0 Then  
'如果該排列組合之特數量超過 1，且自身之目前血量扣除特數量仍大於 0 時  
atking = atking & "EventActiveAIScore#1,2"  
'判斷為「推薦」之排列組合、評分加權為數變化量優先型評分  
For i = 1 To UBound(VBEPageCardNumVS, 1)  
If VBEPageCardNumVS(i, 1) = 4 Then  
'在排列組合的牌中，如果其中類型為特卡時  
atking = atking & "," & i '標記為具貢獻之卡牌  
End If  
Next  
atking = atking & "="  
ElseIf VBEAtkingVSF(1, 1, 1) - AtkingPagetotVS(1, 4) <= 0 Then  
'但如果該排列組合會造成自己「自殺」時  
atking = atking & "EventActiveAIScore#3,2" & "="  
'判斷為「不推薦」之排列組合  
End If  
End If  
End Select  
End Function
```

9. 腳本之錯誤處理

當腳本愈寫愈大之際，難免的裡面一定會有所謂的臭蟲 (Bug) 出現，雖然本系統不能像一般程式設計軟體一樣，具對每一行之逐步偵錯之功能，但我們還是有提供一些錯誤訊息給各位做參考。

當遊戲系統發生錯誤時，在執行階段系統以及執行指令集系統有提供錯誤訊息之顯示，如圖(9-1)。前面 2 行為遊戲系統之錯誤顯示訊息，而最後一行則為 VB6 內部之 Err 錯誤訊息。



圖(9-1)

以下表格為系統之錯誤編號說明：

遊戲系統錯誤編號表

類別	主編號	次編號	詳細說明
執行階段系統 (03)	03-1-1	-	異常狀態類腳本搜尋錯誤
	03-1-2	-	Microsoft Script Control 物件 創立錯誤
	03-1-3	[<技能唯一識別碼>]	主動技能/被動技能類 腳本讀入錯誤(AddCode)
	03-1-3	[<腳本檔案路徑>]	人物實際狀態/異常狀態類 腳本讀入錯誤(AddCode)
	03-1-4	[<1.使用者方/2.電腦方> -<人物技能編號>]	主動技能/被動技能類 初始資料讀入錯誤(初始架構)

遊戲系統錯誤編號表

類別	主編號	次編號	詳細說明
執行階段系統 (03)	03-1-5	[<腳本檔案路徑>]	異常狀態類 初始資料讀入錯誤(初始架構)
	03-1-6	-	人物實際狀態類腳本搜尋錯誤
	03-1-7	[<腳本檔案路徑>]	人物實際狀態類 初始資料讀入錯誤(初始架構)
	03-2-1	[<1.使用者方/2.電腦方> -<執行階段號> -<技能唯一識別碼>]	執行階段系統 執行技能驗證時錯誤
	03-2-2	[1-<人物技能編號>]	主動技能類 執行腳本時錯誤(主架構)
	03-2-2	[2-<人物技能編號>]	被動技能類 執行腳本時錯誤(主架構)
	03-2-2	[3-<技能唯一識別碼>]	異常狀態類 執行腳本時錯誤(主架構)
	03-2-2	[4-<技能唯一識別碼>]	人物實際狀態類 執行腳本時錯誤(主架構)
執行指令集系統 (04)	04	-<執行階段號> -<執行指令名稱> -<執行指令集分階段號>	執行指令集系統 指令執行錯誤
	04	-Run-CommadNotFound [執行指令名稱]-0 -<執行指令集分階段號>	執行指令集系統 未找到該指令之錯誤

以下為截自網路上對 VBScript 之錯誤訊息一覽：

VBScript 錯誤編號表

錯誤代碼	訊息
5	不正確的程序呼叫
6	溢位
7	記憶體不足
9	陣列索引超出範圍
10	此陣列為固定的或暫時鎖定
11	除以零
13	型態不符合
14	字串空間不足

VBScript 錯誤編號表

錯誤代碼	訊息
28	堆疊空間不足
35	Sub 或 Function 沒有定義
48	載入 DLL 時發生錯誤
51	內部錯誤
53	找不到檔案
57	週邊設備 I/O 錯誤
58	檔案已經存在
61	磁碟已滿
67	檔案過多
70	沒有使用權限
75	路徑/檔案存取錯誤
76	找不到路徑
91	尚未設定物件變數或 With 區塊變數
92	For 迴圈沒有初始化
94	不正確使用 Null
322	無法建立必要的暫時檔案
424	此處需要物件
429	ActiveX 元件無法建立物件
430	物件類別不支援 Automation
432	在 Automation 作業期間找不到檔案或物件類別名稱
438	物件不支援此屬性或方法
440	Automation 錯誤
445	物件不支援此動作
446	物件不支援指名引數
447	物件不支援目前的地區設定
448	找不到指名引數
449	引數不為選擇項
450	引數的個數錯誤或不正確的屬性設定
451	物件並非集合物件
453	找不到指定的 DLL 函數
455	程式碼鎖定錯誤
457	此索引鍵已經與集合物件中的某元素相關
458	變數使用的型態是 VBScript 不支援的
500	沒有定義變數

VBScript 錯誤編號表

錯誤代碼	訊息
501	指定不合法
502	物件無法安全地於 script 中使用
503	物件無法安全地初始化
1001	記憶體不足
1002	語法錯誤
1003	必須要有':'
1004	必須要有':'
1005	必須要有'('
1006	必須要有')'
1007	必須要有']'
1008	必須要有'{'
1009	必須要有'}'
1010	必須要有識別項
1011	必須要有'='
1012	必須要有'If'
1013	必須要有'To'
1014	必須要有'End'
1015	必須要有'Function'
1016	必須要有'Sub'
1017	必須要有'Then'
1018	必須要有'Wend'
1019	必須要有'Loop'
1020	必須要有'Next'
1021	必須要有'Case'
1022	必須要有'Select'
1023	必須提供運算式
1024	必須提供陳述式
1025	必須是陳述式的結尾
1026	必須要有整數常數
1027	必須要有'While'或'Until'
1028	必須要有'While'、'Until'或為陳述式的結尾
1029	區域變數或引數過多
1030	識別項太長
1031	不正確的數字
1032	不正確的字元

VBScript 錯誤編號表

錯誤代碼	訊息
1033	無法判定字串常數的結尾
1034	無法判定程式註解的結尾
1035	程式註解形成巢狀
1037	Me'關鍵字的用法不正確
1038	有'loop'，但沒有'do'
1039	exit'陳述式不正確
1040	For'迴圈的計數變數不正確
1041	名稱重複定義
1042	必須是此行的第一個陳述式
1043	無法指定至非 ByVal 型態的引數
1044	呼叫副程式(Sub)時，不能使用括號
1045	必須要有字母(literal)常數
1046	必須要有'ln'
32766	TRUE
32767	FALSE
32811	找不到元素

在腳本中的大部分的語法錯誤都應會在(03-1-3)號錯誤，也就是腳本物件在 AddCode 的時候就會被發現，有些錯誤則會遊戲執行中的時候才會發生。

10. 附錄 1-人物角色之圖片資源概要

本章節將列舉一位人物角色在遊戲系統中所需要之圖片資源，來讓各

位做為製作角色時之參考：

人物角色圖片資源表


(引用:自創人物角色-蕾 (VBEID:S001))

項目	圖片範例	格式
角色卡片		PNG/JPG [須為(168*240)px] (註：因系統已自動標上人物白值，如圖片上含有白值數字者請將白值數字部分去除。)
大人物立繪		PNG
小人物立繪		PNG
小人物立繪影子		PNG (註：影子須加上半透明效果約 50%)


人物角色圖片資源表

(引用:自創人物角色-蕾 (VBEID:S001))

項目	圖片範例	格式
小人物頭像		<p>PNG</p> <p>[須為(248*80)px]</p> <p>(註：於人物頭像外為全透明效果)</p>
各技能圖		<p>PNG</p> <p>(註¹：靜態與逐格動畫皆適用之)</p> <p>(註²：高度 match 為 660px)</p>



2.選擇「系統」標籤



3.選擇「測試模式(開啟自訂影子座標)」



4.在進入戰鬥後，選擇右下方之「影子設定」按鈕



5.可操作 4 向按鈕使影子移動，當與小人物立繪相吻合時即可抄下座標。

L：影子朝左邊移動 10 單位 R：影子朝右邊移動 10 單位
 Tup：影子朝上方移動 10 單位 Tdn：影子朝下方移動 10 單位
 1P：使用者方 COM：電腦方